Maple-assisted proof of formula for A269603

Robert Israel

24 Jan 2019

Let a "state" consist of a pair (x, v) where $x \in [0,..., 5]$ and $v \in [0,..., 6]$. We interpret this as saying that the current value is x and the last repeated value was v (or, if $v = 6$, there was no previous repeated value). We start in any initial state $(x, 6)$. The allowed transitions from (x, v) are to (y, v) if $y \neq x$, or to (x, x) if $|x - v| > 1$. or $v = 6$. Listing the states as $s_i = (x, v)$ where $x = (i - 1) \mathbf{mod}\, 6$ and $i = 1 + x + 6\, v$, $i = 1 \ldots 42$, we have the 42×42 transition matrix constructed below:

```
> with(LinearAlgebra):
  T:= Matrix(42,42):
  for x from 0 to 5 do
    for v from 0 to 6 do
      i:= 1 + x + 6*v;
      for y in {$0..5} minus {x} do
        T[i,1+y+6*v]:= 1;
      od:
      if abs(x-v) > 1 or v=6 then T[i,1+x+6*x]:= 1 fi
  od od:
```

Then $a(n) = \sum_{i=37}^{42} \sum_{j=1}^{42} (T^{n-1})_{ij} = u^T\, T^{n-1}\, v$ where u is the vector with the last 6 entries 1 and the others 0, while v is the vector of all 1's. To check, here are the first few values.

```
> u:= Vector([0$36,1$6]): v:= Vector(42,1):
> Tv[1]:= v: for n from 2 to 30 do Tv[n]:= T . Tv[n-1] od:
  seq(u^%T . Tv[n], n=1..30);
```

6, 36, 210, 1220, 7030, 40288, 229754, 1304934, 7385898, 41679780, 234601902, 1317558578, 7385249086, 41325945826, 230904832646, 1288466651340, 7181415415962, 39985405920156, 222432351559566, 1236355637246456, 6867149321517682, 38118089644078918, 211464390364885802, 1172522774648593308, 6498403478221701606, 36009682382827307770, 199371135622415218590, 1103741621501676864386, 6108636061528359154822, 33799156546350610835860 **(1)**

Here is the empirical recurrence formula. It says that $u^T Q(T)\, T^{n-1}\, v = 0$ for all nonnegative integers n, where Q is the following polynomial.

```
> n:= 'n': empirical:= a(n) = 17*a(n-1) - 91*a(n-2) + 83*a(n-3) +
  542*a(n-4) - 550*a(n-5) - 1651*a(n-6) - 745*a(n-7):
  Q:= unapply(add(coeff((lhs-rhs)(empirical),a(n-i))*t^(7-i),i=0.
  .7),t);
```

$$Q := t \mapsto t^7 - 17\, t^6 + 91\, t^5 - 83\, t^4 - 542\, t^3 + 550\, t^2 + 1651\, t + 745 \qquad \textbf{(2)}$$

In fact, it turns out that $u^T\, Q(T) = 0$, so this is true.

```
> uT[0]:= u^%T:
  for n from 1 to 7 do uT[n]:= uT[n-1] . T od:
  uQ:= add(coeff(Q(t),t,n)*uT[n],n=0..7):
> uQ;
```

[0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, **(3)**

$\lfloor \quad 0, 0, 0, 0, 0, 0, 0, 0, 0, 0]$

This completes the proof.